OFR 71-277

LINEAR COMBINATION READING PROGRAM FOR CAPTURE GAMMA RAYS

by

Allan B. Tanner
U. S. Geological Survey
Washington, D. C. 20242

ABSTRACT

This program computes a weighting function, Q_j, which gives a scalar output value of unity when applied to the spectrum of a desired element and a minimum value (considering statistics) when applied to spectra of materials not containing the desired element. Intermediate values are obtained for materials containing the desired element, in proportion to the amount of the element they contain. The program is written in the BASIC language in a format specific to the Hewlett-Packard 2000A Time-Sharing System, and is an adaptation of an earlier program for linear combination reading for X-ray fluorescence analysis (Tanner and Brinkerhoff, 1971). Following the program is a sample run from a study of the application of the linear combination technique to capture-gamma-ray analysis for calcium (report in preparation).

REFERENCE

Tanner, A. B. and Brinkerhoff, J. M., 1971, Techniques for enhancing sensitivity of X-ray fluorescence analysis of rocks, in Ziegler, C. A., ed., Applications of low energy X- and gamma rays: New York, Gordon and Breach Science Publishers.

```
10 REM  *LINEAR COMBINATION READING PROGRAM FOR CAPTURE GAMMA RAYS*
12 REM  THIS PROGRAM COMPUTES A WEIGHTING FUNCTION, Q[J], WHICH
14 REM  GIVES A SCALAR OUTPUT VALUE OF UNITY WHEN APPLIED TO THE
16 REM  SPECTRUM OF A DESIRED ELEMENT AND A MINIMUM VALUE (CONSIDERING
18 REM  STATISTICS) WHEN APPLIED TO SPECTRA OF MATERIALS NOT CONTAINING
20 REM  THE DESIRED ELEMENT.  INTERMEDIATE VALUES ARE OBTAINED FOR
22 REM  MATERIALS CONTAINING THE DESIRED ELEMENT, IN PROPORTION TO THE
24 REM  AMOUNT OF THE ELEMENT THEY CONTAIN.
26 REM       H2 "WINDOWS" ARE CHOSEN TO BRACKET PROMINENT PEAKS IN THE
28 REM  SPECTRUM OF THE DESIRED ELEMENT; INFORMATION WITHIN AND ABOUT
30 REM  EACH WINDOW IS SUPPLIED AS A TOTAL OF H1 DATA CHANNELS.  FOR
32 REM  EACH WINDOW THERE ARE, THEN, H1/H2 DATA CHANNELS.  FROM THE
34 REM  DATA CHANNELS CORRESPONDING TO EACH WINDOW EQUAL GROUPS ARE
36 REM  MADE UP, H3 CHANNELS PER GROUP, SYMMETRICALLY SPACED ABOUT THE
38 REM  MIDPOINT OF THE WINDOW.  AS THERE ARE N TOTAL GROUPS THAT MAY
40 REM  BE USED IN THE COMPUTATION, (N/H2)*H3 DATA CHANNELS ARE USED
42 REM  FROM THE MIDDLE OF EACH SET OF DATA CHANNELS.  SPECTRA FROM
44 REM  M BACKGROUND MATERIALS, ACCUMULATED FOR TIME T1, ARE USED IN
46 REM  THE COMPUTATION; THE RESULTING WEIGHTING FUNCTION Q IS TO BE
48 REM  APPLIED TO SPECTRA ACCUMULATED IN TIME T2.  THE SPECTRUM
50 REM  SEGMENTS OF THE DESIRED ELEMENT ARE DENOTED AS VECTOR A;
52 REM  THOSE OF THE BACKGROUND MATERIALS ARE VECTORS V.  "UNKNOWN"
54 REM  SPECTRA MAY THEN BE EVALUATED AS ADDITIONAL VECTORS V.  S$
56 REM  CONTAINS 5 CHARACTERS, THE FIRST DENOTING THE SPECIFIC
58 REM  IRRADIATION CONDITIONS, THE SECOND THE SET OF MATERIALS USED
60 REM  FOR THE COMPUTATION, THE THIRD A FILLER, THE FOURTH THE UNIQUE
62 REM  COMBINATION OF PARAMETERS USED, AND THE FIFTH DESIGNATING
64 REM  WHETHER ADDITIONAL VECTORS ARE TO BE EVALUATED: "-" IF NOT,
66 REM  "+" IF SO, AND A LETTER OR NUMBER IF THE COMPUTATION IS ONLY
68 REM  FOR EVALUATING ADDITIONAL VECTORS USING THE Q FUNCTION STORED
70 REM  IN DATA FILE LCF.  DATA FILES MUST BE ESTABLISHED PRIOR TO
72 REM  RUNNING THE PROGRAM.THE PROGRAM FORMAT IS SPECIFIC FOR THE
74 REM  HEWLETT-PACKARD 2000A BASIC TIME-SHARING SYSTEM.
100   DATA "F5-5+",10,24,216,6,5,16384,16384
200   DATA "F5-5A",7
1000   FILES LCF,LCD,LCE
1005   DIM B[216],U[216]
1010   DIM A[25],C[25,25],P[25],Q[25],S$[5]
1020   DIM E[10],F[10],V[10,24]
1030   READ S$,M,N,H1,H2,H3,T1,T2
1040   IF S$[5,5]="-" THEN 1060
1050   IF S$[5,5]#"+" THEN 2160
1060   MAT  READ #2;B
1070   MAT A=ZER[N+1]
1090   LET H4=N/H2
1094   FOR L=0 TO H2-1
1096   FOR J=L*H4+1 TO (L+1)*H4
1098   FOR K=1 TO H3
1100   LET A[J]=A[J]+B[L*H1/H2+(H1-N*H3)/(H2*2)+H3*(J-4*L-1)+K]
1102   NEXT K
1104   NEXT J
1106   NEXT L
1120   MAT A=(T2/T1)*A
1121   PRINT "GROUPED AND TIME-CORRECTED VALUES OF VECTOR A"
1123   FOR J=1 TO N
1124   PRINT A[J];
```

```
1126  NEXT J
1128  PRINT
1129  PRINT
1130  MAT Q=ZER[N+1]
1140  LET Q[N+1]=1
1150  MAT C=ZER[N+1,N+1]
1160  MAT P=ZER[N+1]
1180  LET R1=R2=Y1=Y2=0
1205  PRINT "GROUPED AND TIME-CORRECTED VALUES OF CALIBRATION VECTORS"
1210  GOSUB 2890
1230  REM  CALCULATION OF MATRIX OF COEFFICIENTS OF QS
1240  FOR K=1 TO N
1250  FOR L=1 TO N
1260  IF K=L THEN 1330
1270  LET C[K,L]=0
1280  FOR J=1 TO M
1290  LET C[K,L]=C[K,L]+V[J,K]*V[J,L]
1300  NEXT J
1310  GOTO 1370
1320  REM  FOR DIAGONAL TERMS
1330  LET C[K,L]=0
1340  FOR J=1 TO M
1350  LET C[K,L]=C[K,L]+V[J,K]+V[J,K]↑2
1360  NEXT J
1370  NEXT L
1380  NEXT K
1390  REM  COMPLETE MATRIX OF COEFFICIENTS
1400  FOR J=1 TO N
1410  LET C[J,N+1]=A[J]
1420  LET C[N+1,J]=A[J]
1430  NEXT J
1440  GOSUB 1650
1450  MAT  PRINT #1;Q
1460  GOTO 2040
1470  REM  ROW INTERCHANGE
1480  LET X1=-1
1490  FOR K=X3 TO N+1
1500  IF ABS(C[K,X3]) <= X1 THEN 1530
1510  LET X2=K
1520  LET X1=ABS(C[K,X3])
1530  NEXT K
1540  REM  START INTERCHANGE, IF NEEDED
1550  IF X3=X2 THEN 1640
1560  FOR J=1 TO N+1
1570  LET X4=C[X3,J]
1580  LET C[X3,J]=C[X2,J]
1590  LET C[X2,J]=X4
1600  NEXT J
1610  LET X4=Q[X3]
1620  LET Q[X3]=Q[X2]
1630  LET Q[X2]=X4
1640  RETURN
1650  LET X3=1
1660  GOSUB 1470
1670  IF C[1,1]=0 THEN 2030
1680  FOR J=2 TO N+1
1690  LET C[1,J]=C[1,J]/C[1,1]
1700  NEXT J
1710  LET Q[1]=Q[1]/C[1,1]
1720  FOR X3=2 TO N+1
```

```
1730  GOSUB 1480
1740  FOR I=X3 TO N+1
1750  LET X5=0
1760  FOR K=1 TO X3-1
1770  LET X5=X5+C[I,K]*C[K,X3]
1780  NEXT K
1790  LET C[I,X3]=C[I,X3]-X5
1800  NEXT I
1810  IF C[X3,X3]=0 THEN 2030
1820  FOR J=X3+1 TO N+1
1830  LET X5=0
1840  FOR K=1 TO X3-1
1850  LET X5=X5+C[X3,K]*C[K,J]
1860  NEXT K
1870  LET C[X3,J]=(C[X3,J]-X5)/C[X3,X3]
1880  NEXT J
1890  LET X5=0
1900  FOR K=1 TO X3-1
1910  LET X5=X5+C[X3,K]*Q[K]
1920  NEXT K
1930  LET Q[X3]=(Q[X3]-X5)/C[X3,X3]
1940  NEXT X3
1950  FOR I=N TO 1 STEP -1
1960  LET X5=0
1970  FOR K=I+1 TO N+1
1980  LET X5=X5+C[I,K]*Q[K]
1990  NEXT K
2000  LET Q[I]=Q[I]-X5
2010  NEXT I
2020  RETURN
2030  PRINT "MATRIX OF COEFFICIENTS IS SINGULAR"
2035  STOP
2040  REM  COMPUTATION OF ERRORS
2045  GOSUB 2520
2050  GOSUB 2290
2060  GOSUB 2380
2070  PRINT "SERIES&NO ";S$;TAB(18);"CALIB. SECS";T1;
2030  PRINT TAB(43)"STD. SECS";T2;TAB(64)"ZERO BKG"
2085  PRINT H3;"DATA CHANNELS PER GROUP"
2090  PRINT
2120  GOSUB 2640
2130  GOSUB 2750
2140  GOSUB 2830
2150  IF S$[5,5]="-" THEN 2280
2160  IF S$[5,5]="+" THEN 2210
2170  MAT Q=ZER[N+1]
2180  MAT  READ #1,1;Q
2190  READ #1;R3
2200  GOTO 2220
2210  READ S$,M
2220  PRINT "GROUPED AND TIME-CORRECTED VALUES OF ADDITIONAL VECTORS"
2230  GOSUB 2886
2240  PRINT "TEST OF COMPUTED QS ON ADDITIONAL VECTORS, RUN ";S$
2250  PRINT
2260  GOSUB 2290
2270  GOSUB 2750
2280  STOP
2290  MAT E=ZER[M]
2300  MAT F=ZER[M]
2310  FOR I=1 TO M
```

```
2320  FOR J=1 TO N
2330  LET E[I]=E[I]+Q[J]*V[I,J]
2340  LET F[I]=F[I]+(Q[J]↑2)*V[I,J]
2345  NEXT J
2360  NEXT I
2370  RETURN
2380  FOR I=1 TO M
2390  LET R1=R1+E[I]↑2
2400  LET R2=R2+F[I]
2410  NEXT I
2420  LET R3=SQR((R1+R2)/M)
2430  LET R4=INT((1/R3)+.5)
2440  PRINT #1;R3
2450  LET R1=SQR(R1/M)
2460  LET R2=SQR(R2/M)
2465  RETURN
2520  FOR J=1 TO N+1
2530  LET Y1=Y1+Q[J]*A[J]
2540  LET P[J]=Y1
2550  NEXT J
2630  RETURN
2640  PRINT TAB(10)"CHANNEL  WEIGHT  [Q]";TAB(37)"CUMULATIVE Q*A"
2650  PRINT TAB(21);"(X1.E+07)"
2660  PRINT
2670  FOR J=1 TO N
2690  PRINT TAB(12);J;TAB(18);INT(1.E+07*Q[J]+.5);
2695  PRINT TAB(40);INT(100000.*P[J]+.5)/100000.
2700  NEXT J
2720  PRINT
2730  RETURN
2750  PRINT " V";TAB(6);"SYSTEMATIC DEPARTURE";TAB(29);
2760  PRINT "STATISTICAL ERROR";TAB(51);"S/N"
2770  FOR I=1 TO M
2780  PRINT " ";I;TAB(11);INT(100000.*E[I]+.5)/100000.;TAB(33);
2781  PRINT INT(100000.*SQR(F[I])+.5)/100000.;TAB(50);
2790  PRINT INT((100*E[I]/R3)+.5)/100;
2795  PRINT
2800  NEXT I
2810  PRINT
2820  RETURN
2830  PRINT "RMS SYSTEMATIC ERROR";TAB(23);"RMS STATISTICAL ERROR";
2840  PRINT TAB(46);"TOTAL ERROR";TAB(61);"S/N"
2850  PRINT TAB(6);INT(10000*R1+.5)/10000;TAB(30);
2851  PRINT INT(10000*R2+.5)/10000;TAB(48);INT(10000*R3+.5)/10000;
2852  PRINT TAB(55);R4
2860  PRINT
2870  RETURN
2886  LET Z=3
2888  GOTO 2892
2890  LET Z=2
2892  MAT V=ZER[M,N]
2894  FOR I=1 TO M
2898  MAT  READ #Z;U
2902  FOR L=0 TO H2-1
2904  FOR J=L*H4+1 TO (L+1)*H4
2906  FOR K=1 TO H3
2910  LET V[I,J]=V[I,J]+U[L*H1/H2+(H1-N*H3)/(H2*2)+H3*(J-4*L-1)+K]
2930  NEXT K
2940  NEXT J
2950  NEXT L
```

```
2960  NEXT I
2970  MAT V=(T2/T1)*V
2972  PRINT "I","FIRST";H4;"TERMS OF V[I,J]"
2974  FOR I=1 TO M
2976  PRINT I;
2978  FOR J=1 TO H4
2980  PRINT V[I,J];
2982  NEXT J
2984  PRINT
2986  NEXT I
2987  PRINT
2988  RETURN
4000  END
```

Sample Run

```
      SAMPLE OF OUTPUT INFORMATION FROM COMPUTER RUN
   RUN
  LCN7     1524

A  GROUPED AND TIME-CORRECTED VALUES OF VECTOR A
   1955      2727      2359      1946      1441      1502      1513      1436
   852    1052      1048      938    848    1445      2876      1665      826
   1130      539    469    206    416    461    144

B  GROUPED AND TIME-CORRECTED VALUES OF CALIBRATION VECTORS
   I                FIRST 4    TERMS OF V[I,J]
    1      10119      10089      10610      10716
    2      6228       6223       6264       6213
    3      2663       2837       2735       2768
    4      17649      17574      17547      18179
    5      2468       2472       2332       2301
    6      6926       6727       6861       6522
    7      2366       2373       2268       2375
    8      1946       2091       2381       2294
    9      1276       1206       1270       1327
    10     4784       4872       4882       4960

C  SERIES&NO F5-5+   CALIB. SECS 16384        STD. SECS 16384       ZERO BKG
   5    DATA CHANNELS PER GROUP

            CHANNEL   WEIGHT [Q]        CUMULATIVE Q*A
                       (X1.E+07)

                1     -2824                -.55205
                2      6171                 1.13075
                3      1943                 1.58919
                4     -1826                 1.23387
                5     -1940                 .95438
                6     -115                  .93716
                7     -926                  .797
                8     -1971                 .51395
                9     -540                  .46794
                10     2752                 .75749
                11    -925                  .66057
                12    -1286                 .53996
                13     41                   .54343
                14    -2190                 .22703
                15     75                   .24853
                16     6233                 1.28632
                17    -1372                 1.17302
                18    -410                  1.12666
                19    -3497                 .93817
                20    -4107                 .74556
                21     1859                 .78386
                22     1690.                .85415
                23     2974                 .99127
                24     606                  1
```

D

V	SYSTEMATIC DEPARTURE	STATISTICAL ERROR	S/N
1	-.00722	.1024	-.09
2	.00145	.08683	.02
3	.00532	.0484	.07
4	-.00092	.14621	-.01
5	.00488	.0451	.06
6	.01437	.07238	.18
7	-.00685	.04517	-.09
8	.00314	.04191	.04
9	.02938	.03148	.37
10	.0138	.08811	.17

E

RMS SYSTEMATIC ERROR	RMS STATISTICAL ERROR	TOTAL ERROR	S/N
.0119	.0785	.0794	13

F GROUPED AND TIME-CORRECTED VALUES OF ADDITIONAL VECTORS

I	FIRST 4	TERMS OF V[I,J]		
1	4686	4617	4745	4655
2	4521	4493	4551	4812
3	7707	7426	7435	7641
4	2600	2791	2827	2313
5	2064	2323	2198	2021
6	2053	2199	2138	1965
7	2084	2117	2036	2080

G TEST OF COMPUTED QS ON ADDITIONAL VECTORS, RUN F5-5A

V	SYSTEMATIC DEPARTURE	STATISTICAL ERROR	S/N
1	.44208	.07192	5.57
2	-.51482	.06413	-6.49
3	.45568	.10059	5.74
4*	1.03167	.05622	13
5	.50243	.04412	6.33
6	.38491	.04495	4.85
7	.06668	.0428	.84

* SPECTRUM RUN UNDER NONSTANDARD CONDITIONS; NOT COMPARABLE WITH OTHER DATA.